Homogene lineare Differentialgleichungen

Übungsaufgabe 1

Wir betrachten die Differentialgleichung

In[16]:=
```
DE = y'''[x] + 3 y''[x] + 3 y'[x] + 2 y[x] == 0
```

Out[16]= $y^{(3)}(x)+3\,y''(x)+3\,y'(x)+2\,y(x)=0$

mit dem charakteristischen Polynom

In[17]:=
```
charpol = DE /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```
Ableitung

Out[17]= $\lambda^3+3\,\lambda^2+3\,\lambda+2=0$

Das charakteristische Polynom hat die komplexen Lösungen:

In[18]:=
```
lösung = Solve[charpol]
```
löse

Out[18]= $\left\{\{\lambda\to-2\},\left\{\lambda\to-\sqrt[3]{-1}\right\},\left\{\lambda\to(-1)^{2/3}\right\}\right\}$

In[19]:=
```
lösung = MapAll[ComplexExpand, lösung]
```
wende ⋯ erweitere komplexen Ausdruck

Out[19]= $\left\{\{\lambda\to-2\},\left\{\lambda\to-\frac{1}{2}-\frac{i\sqrt{3}}{2}\right\},\left\{\lambda\to-\frac{1}{2}+\frac{i\sqrt{3}}{2}\right\}\right\}$

Entsprechend erhalten wir die Lösungsbasis der Differentialgleichung DE:

In[20]:=
```
komplexebasis = Map[Exp[# x] &, λ /. lösung]
```
w⋯ Exponentialfunktion

Out[20]= $\left\{e^{-2\,x},\,e^{\left(-\frac{1}{2}-\frac{i\sqrt{3}}{2}\right)x},\,e^{\left(-\frac{1}{2}+\frac{i\sqrt{3}}{2}\right)x}\right\}$

In[21]:=
```
reellebasis =
  Union[ComplexExpand[Re[komplexebasis]], ComplexExpand[Im[komplexebasis]]]
```
Verein⋯ erweitere komplex⋯ Realteil erweitere komplex⋯ Imaginärteil

Out[21]= $\left\{0,\,e^{-2\,x},\,e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right),\,-e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right),\,e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right)\right\}$

In[22]:=
```
reellebasis = {reellebasis[[2]], reellebasis[[3]], reellebasis[[5]]}
```

Out[22]= $\left\{e^{-2\,x},\,e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right),\,e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right)\right\}$

DSolve liefert:

In[28]:=
```
sol = DSolve[DE, y[x], x]
```
löse Differentialgleichung

Out[28]= $\left\{\left\{y(x)\to c_3\, e^{-2x}+c_1\, e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right)+c_2\, e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)\right\}\right\}$

Wir lösen nun ein Anfangswertproblem mit y(0)=3, y'(0)=0, y''(0)=2:

In[24]:=
```
WronskiMatrix[basis_, x_] :=
 Table[D[basis[[k]], {x, j}], {j, 0, Length[basis] - 1}, {k, Length[basis]}]
```
Tabelle leite ab Länge Länge

```
WronskiDeterminante[basis_, x_] := Det[WronskiMatrix[basis, x]]
```
Determinante

In[31]:=
```
gl1 = (y[x] == 3 /. sol[[1]]) /. x → 0
```

Out[31]= $c_2+c_3=3$

In[34]:=
```
gl2 = (D[y[x] /. sol[[1]], x] == 0) /. x → 0
```
leite ab

Out[34]= $\frac{\sqrt{3}\,c_1}{2}-\frac{c_2}{2}-2\,c_3=0$

In[35]:=
```
gl3 = (D[y[x] /. sol[[1]], {x, 2}] == 2) /. x → 0
```
leite ab

Out[35]= $-\frac{1}{2}\sqrt{3}\,c_1-\frac{c_2}{2}+4\,c_3=2$

In[36]:=
```
Solve[{gl1, gl2, gl3}, {C[1], C[2], C[3]}]
```
löse Kons··· Kons··· Konstante

Out[36]= $\left\{\left\{c_1\to\frac{8}{\sqrt{3}},\ c_2\to\frac{4}{3},\ c_3\to\frac{5}{3}\right\}\right\}$

In[37]:=
```
DSolve[y''[x] == x * y[x], y[x], x]
```
löse Differentialgleichung

Out[37]= $\{\{y(x)\to c_1\,\mathrm{Ai}(x)+c_2\,\mathrm{Bi}(x)\}\}$

In[38]:=
```
DSolve[y'''[x] == x * y[x], y[x], x]
```
löse Differentialgleichung

Out[38]= $\left\{\left\{y(x)\to\frac{\sqrt[4]{-1}\,c_2\,x\,{}_0F_2\left(;\frac{3}{4},\frac{5}{4};\frac{x^4}{64}\right)}{2\sqrt{2}}+c_1\,{}_0F_2\left(;\frac{1}{2},\frac{3}{4};\frac{x^4}{64}\right)+\frac{1}{8}\,i\,c_3\,x^2\,{}_0F_2\left(;\frac{5}{4},\frac{3}{2};\frac{x^4}{64}\right)\right\}\right\}$

Inhomogene lineare Differentialgleichungen

Beispiel 4.6, Übungsaufgabe 3

In[57]:=
```
Clear[u]
```
lösche

In[58]:=
```
DE = y''[x] + 4 y'[x] + 4 y[x]
```

Out[58]= $y''(x) + 4\, y'(x) + 4\, y(x)$

homogene Gleichung

In[59]:=
```
charpol = DE /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```
Ableitung

Out[59]= $\lambda^2 + 4\,\lambda + 4$

In[60]:=
```
Solve [charpol == 0, λ]
```
löse

Out[60]= $\{\{\lambda \to -2\}, \{\lambda \to -2\}\}$

In[61]:=
```
DSolve[DE == 0, y[x], x]
```
löse Differentialgleichung

Out[61]= $\{\{y(x) \to c_1\, e^{-2x} + c_2\, e^{-2x}\, x\}\}$

Iterative Methode: Wir lösen (nach Faktorisierung) zuerst eine inhomogene Differentialgleichung erster Ordnung:

In[62]:=
```
lösung = DSolve[u'[x] + 2 u[x] == E^(-2 x)/x^2, u[x], x]
```
löse Differentialgleichung

Out[62]= $\left\{\left\{u(x) \to c_1\, e^{-2x} - \frac{e^{-2x}}{x}\right\}\right\}$

In[63]:=
```
U[x_] = Evaluate[u[x] /. lösung[[1]] /. C[1] → 0]
```
werte aus; Konstante

Out[63]= $-\frac{e^{-2x}}{x}$

und finden im zweiten Schritt eine spezielle Lösung der inhomogenen Differentialgleichung:

In[64]:=
```
DSolve[y'[x] + 2 y[x] == U[x], y[x], x]
```
löse Differentialgleichung

Out[64]= $\{\{y(x) \to c_1\, e^{-2x} - e^{-2x}\, \log(x)\}\}$

Vergleich: DSolve

In[65]:= `DSolve[DE == e^(-2 x)/x^2, y[x], x]`
löse Differentialgleichung

Out[65]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-2x} x - e^{-2x} (\log(x) + 1)\right\}\right\}$

Ansatzmethode

Beispiel (a):

In[66]:= `DE = y''''[x] - 5 y''[x] + 4 y[x]`

Out[66]= $y^{(4)}(x) - 5\, y''(x) + 4\, y(x)$

homogene Gleichung

In[67]:= `DSolve[DE == 0, y[x], x]`
löse Differentialgleichung

Out[67]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-x} + c_3 e^{x} + c_4 e^{2x}\right\}\right\}$

inhomogene Gleichung

In[68]:= `Lösung = DSolve[DE == 1 + x + x^2, y[x], x]`
löse Differentialgleichung

Out[68]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-x} + c_3 e^{x} + c_4 e^{2x} + \frac{1}{8}\left(2x^2 + 2x + 7\right)\right\}\right\}$

Wir verwenden den Ansatz

In[69]:= `ansatz = a x^2 + b x + c`

Out[69]= $a x^2 + b x + c$

Bevor wir ihn einsetzen, schreiben wir den Ansatz als Funktion:

In[70]:= `Function[{x}, Evaluate[ansatz]]`
Funktion werte aus

Out[70]= $\{x\} \longmapsto a x^2 + b x + c$

und setzen dies in die inhomogene Differentialgleichung ein:

In[71]:= `ausdruck = DE - (1 + x + x^2) /. {y → Function[{x}, Evaluate[ansatz]]}`
Funktion werte aus

Out[71]= $4\left(a x^2 + b x + c\right) - 10\, a - x^2 - x - 1$

Koeffizientenvergleich liefert

In[72]:=
```
lösung = Solve[CoefficientList[ausdruck, x] == 0, {a, b, c}]
```
löse   Liste der Koeffizienten

Out[72]= $\left\{\left\{a \to \frac{1}{4},\, b \to \frac{1}{4},\, c \to \frac{7}{8}\right\}\right\}$

Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:

In[73]:=
```
ansatz /. lösung[[1]]
```

Out[73]= $\frac{x^2}{4} + \frac{x}{4} + \frac{7}{8}$

Vergleich:

In[74]:=
```
Lösung
```

Out[74]= $\left\{\left\{y(x) \to c_1\, e^{-2x} + c_2\, e^{-x} + c_3\, e^{x} + c_4\, e^{2x} + \frac{1}{8}\left(2x^2 + 2x + 7\right)\right\}\right\}$

(b)

In[75]:=
```
DE = y'''[x] - y'[x]
```

Out[75]= $y^{(3)}(x) - y'(x)$

homogene Gleichung

In[76]:=
```
DSolve[DE == 0, y[x], x]
```
löse Differentialgleichung

Out[76]= $\{\{y(x) \to c_1\, e^{x} - c_2\, e^{-x} + c_3\}\}$

inhomogene Gleichung

In[77]:=
```
Lösung = DSolve[DE == E^x + E^(2 x), y[x], x]
```
löse Differentialgleichung

Out[77]= $\left\{\left\{y(x) \to e^{x}\left(\frac{1}{4}(4c_1 - 3) + \frac{x}{2}\right) - c_2\, e^{-x} + c_3 + \frac{e^{2x}}{6}\right\}\right\}$

Wir verwenden den Ansatz

In[78]:=
```
ansatz = a E^(2 x) + b x E^x
```

Out[78]= $a\, e^{2x} + b\, e^{x}\, x$

und setzen dies in die inhomogene Differentialgleichung ein:

In[79]:=
```
ausdruck = DE - (E^x + E^(2 x)) /. {y → Function[{x}, Evaluate[ansatz]]}
```
Funktion   werte aus

Out[79]= $6\,a\,e^{2x} + 2\,b\,e^{x} - e^{x} - e^{2x}$

Koeffizientenvergleich liefert

In[80]:=
```
lösung = Solve[{Coefficient[ausdruck, E^x], Coefficient[ausdruck, E^(2 x)]} == 0, {a, b}]
```
löse   Koeffizient   Koeffizient

Out[80]= $\left\{\left\{a \to \frac{1}{6},\, b \to \frac{1}{2}\right\}\right\}$

Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:

In[81]:=
```
ansatz /. lösung[[1]]
```

Out[81]= $\frac{e^{x}\,x}{2} + \frac{e^{2x}}{6}$

Vergleich:

In[82]:=
```
Lösung
```

Out[82]= $\left\{\left\{y(x) \to e^{x}\left(\frac{1}{4}(4\,c_1 - 3) + \frac{x}{2}\right) - c_2\,e^{-x} + c_3 + \frac{e^{2x}}{6}\right\}\right\}$

(c)

In[83]:=
```
DE = y''[x] - 3 y'[x] + 2 y[x]
```

Out[83]= $y''(x) - 3\,y'(x) + 2\,y(x)$

homogene Gleichung

In[84]:=
```
DSolve[DE == 0, y[x], x]
```
löse Differentialgleichung

Out[84]= $\left\{\left\{y(x) \to c_1\,e^{x} + c_2\,e^{2x}\right\}\right\}$

inhomogene Gleichung

In[85]:=
```
Lösung = DSolve[DE == Sin[x], y[x], x]
```
löse Differentia·· Sinus

Out[85]= $\left\{\left\{y(x) \to c_1\,e^{x} + c_2\,e^{2x} + \frac{1}{10}(\sin(x) + 3\cos(x))\right\}\right\}$

Wir verwenden den Ansatz

In[86]:=
```
ansatz = a Sin[x] + b Cos[x]
```
Sinus Kosinus

Out[86]= $a \sin(x) + b \cos(x)$

und setzen dies in die inhomogene Differentialgleichung ein:

In[87]:=
```
ausdruck = DE - (Sin[x]) /. {y → Function[{x}, Evaluate[ansatz]]}
```
Sinus Funktion werte aus

Out[87]= $2\,(a \sin(x) + b \cos(x)) - 3\,(a \cos(x) - b \sin(x)) - a \sin(x) - b \cos(x) - \sin(x)$

Koeffizientenvergleich liefert

In[88]:=
```
lösung =
 Solve[{Coefficient[ausdruck, Sin[x]], Coefficient[ausdruck, Cos[x]]} == 0, {a, b}]
```
löse Koeffizient Sinus Koeffizient Kosinus

Out[88]= $\left\{\left\{a \to \frac{1}{10}, b \to \frac{3}{10}\right\}\right\}$

Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:

In[89]:=
```
ansatz /. lösung[[1]]
```

Out[89]= $\frac{\sin(x)}{10} + \frac{3 \cos(x)}{10}$

Vergleich:

In[90]:=
```
Lösung
```

Out[90]= $\left\{\left\{y(x) \to c_1 e^x + c_2 e^{2x} + \frac{1}{10}(\sin(x) + 3 \cos(x))\right\}\right\}$

Hausaufgabe: Ansatzverfahren für Differentialgleichung

In[91]:=
```
DE = y'''[x] - y[x] == e^x + e^(2 x)
```

Out[91]= $y^{(3)}(x) - y(x) = e^x + e^{2x}$

charakteristisches Polynom

In[92]:=
```
charpol = DE[[1]] /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```
Ableitung

Out[92]= $\lambda^3 - 1$

In[93]:=
```
Factor[charpol]
```
faktorisiere

Out[93]= $(\lambda - 1)(\lambda^2 + \lambda + 1)$

homogene Gleichung

In[94]:=
```
DSolve[DE[[1]] == 0, y[x], x]
```
löse Differentialgleichung

Out[94]= $\left\{\left\{y(x) \to c_1 e^x + c_3 e^{-x/2} \sin\left(\frac{\sqrt{3}\, x}{2}\right) + c_2 e^{-x/2} \cos\left(\frac{\sqrt{3}\, x}{2}\right)\right\}\right\}$

Wir verwenden den Ansatz

In[95]:=
```
ansatz = (a + b x) e^x + c e^(2 x)
```

Out[95]= $e^x (a + b x) + c e^{2x}$

Wir setzen dies in die inhomogene Differentialgleichung ein:

In[96]:=
```
ausdruck =
  Expand[DE[[1]] - DE[[2]] /. Table[D[y[x], {x, k}] → D[ansatz, {x, k}], {k, 0, 3}]]
```
multipliziere aus   Tabelle   leite ab   leite ab

Out[96]= $3 b e^x + 7 c e^{2x} - e^x - e^{2x}$

Koeffizientenvergleich liefert

In[97]:=
```
lösung = Solve[Join[CoefficientList[Coefficient[ausdruck, e^x], x],
    CoefficientList[Coefficient[ausdruck, e^(2 x)], x]] == 0, {a, b, c}]
```
löse   verk···   Liste der Koeffizienten   Koeffizient
Liste der Koeffizienten   Koeffizient

Out[97]= $\left\{\left\{b \to \frac{1}{3},\, c \to \frac{1}{7}\right\}\right\}$

Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:

In[98]:=
```
ansatz /. lösung[[1]]
```

Out[98]= $e^x \left(a + \frac{x}{3}\right) + \frac{e^{2x}}{7}$

Vergleich:

In[99]:=
```
dsol = DSolve[DE, y[x], x]
```
löse Differentialgleichung

Out[99]=
$$\left\{\left\{y(x)\to c_1 e^x + c_3 e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right) + c_2 e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right) - \frac{1}{21} e^x\left(-7 e^x - 7x + 4 e^x \sin^2\left(\frac{\sqrt{3}\,x}{2}\right) + 7\sin^2\left(\frac{\sqrt{3}\,x}{2}\right) + 4 e^x\cos^2\left(\frac{\sqrt{3}\,x}{2}\right) + 7\cos^2\left(\frac{\sqrt{3}\,x}{2}\right)\right)\right\}\right\}$$

In[100]:=
```
Simplify[dsol]
```
vereinfache

Out[100]=
$$\left\{\left\{y(x)\to \frac{1}{21} e^{-x/2}\left(e^{3x/2}(21 c_1 + 7x + 3 e^x - 7) + 21 c_3 \sin\left(\frac{\sqrt{3}\,x}{2}\right) + 21 c_2\cos\left(\frac{\sqrt{3}\,x}{2}\right)\right)\right\}\right\}$$

Differentialgleichungssysteme

Beispiel 3.1

In[101]:=
```
dgl = {y1'[x] == -2 y1[x] + y2[x], y2'[x] == -2 y2[x] + Exp[-2 x]/x}
```

Out[101]=
$$\left\{\mathrm{y1}'(x) = \mathrm{y2}(x) - 2\,\mathrm{y1}(x),\ \mathrm{y2}'(x) = \frac{e^{-2x}}{x} - 2\,\mathrm{y2}(x)\right\}$$

In[102]:=
```
lösung = DSolve[dgl, {y1[x], y2[x]}, x]
```
löse Differentialgleichung

Out[102]=
$$\left\{\left\{\mathrm{y1}(x)\to c_2 e^{-2x} x + c_1 e^{-2x} - e^{-2x} x + e^{-2x} x\log(x),\ \mathrm{y2}(x)\to c_2 e^{-2x} + e^{-2x}\log(x)\right\}\right\}$$

schrittweise:

In[103]:=
```
lösung2 = DSolve[dgl[[2]], y2[x], x]
```
löse Differentialgleichung

Out[103]=
$$\left\{\left\{\mathrm{y2}(x)\to c_1 e^{-2x} + e^{-2x}\log(x)\right\}\right\}$$

In[104]:=
```
dgl[[1]] /. lösung2[[1]]
```

Out[104]=
$$\mathrm{y1}'(x) = c_1 e^{-2x} - 2\,\mathrm{y1}(x) + e^{-2x}\log(x)$$

In[105]:=
```
DSolve [dgl[[1]] /. lösung2[[1]], y1[x], x]
```
löse Differentialgleichung

Out[105]=
$$\left\{\left\{\mathrm{y1}(x)\to c_2 e^{-2x} + e^{-2x}(c_1 x - x + x\log(x))\right\}\right\}$$

Vergleich: